

PhysRevD.67.084023 equation (27)

```
1  from shared import *
2  import cdblib
3
4  jsonfile = 'gamma.json'
5  cdblib.create (jsonfile)
6
7  # -----
8  # Gamma in terms of GammaBar and phi, see prd67 eqn 27
9
10 Gamma := \Gamma^{a}_{b c}. # cdb (eq27.101,Gamma)
11
12 substitute (Gamma, defGamma) # cdb (eq27.102,Gamma)
13 substitute (Gamma, defG2GBarD) # cdb (eq27.103,Gamma)
14 substitute (Gamma, defG2GBarU) # cdb (eq27.104,Gamma)
15 distribute (Gamma) # cdb (eq27.105,Gamma)
16 product_rule (Gamma) # cdb (eq27.106,Gamma)
17 substitute (Gamma, dexp) # cdb (eq27.107,Gamma)
18 distribute (Gamma) # cdb (eq27.108,Gamma)
19 map_sympy (Gamma, "simplify") # cdb (eq27.109,Gamma)
20
21 foo := gBar^{a e} \partial_{e}{gBar_{b c}} ->
22     - 2 GammaBar^{a}_{b c}
23     + gBar^{a e} \partial_{b}{gBar_{e c}}
24     + gBar^{a e} \partial_{c}{gBar_{b e}}.
25
26 substitute (Gamma, foo) # cdb (eq27.110,Gamma)
27 substitute (Gamma, $gBar^{a i} gBar_{i b} -> gBar^{a}_{b}$) # cdb (eq27.111,Gamma)
28 substitute (Gamma, $gBar^{a i} gBar_{b i} -> gBar^{a}_{b}$) # cdb (eq27.112,Gamma)
29
30 defG2GBar := \Gamma^{a}_{b c} -> @(Gamma).
31
32 cdblib.put ('defG2GBar',defG2GBar,jsonfile)
```

$$\Gamma^a_{bc} = \frac{1}{2} g^{ae} (\partial_b g_{ec} + \partial_c g_{be} - \partial_e g_{bc}) \quad (\text{eq27.102})$$

$$= \frac{1}{2} g^{ae} (\partial_b (\exp(4\phi) \bar{g}_{ec}) + \partial_c (\exp(4\phi) \bar{g}_{be}) - \partial_e (\exp(4\phi) \bar{g}_{bc})) \quad (\text{eq27.103})$$

$$= \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} (\partial_b (\exp(4\phi) \bar{g}_{ec}) + \partial_c (\exp(4\phi) \bar{g}_{be}) - \partial_e (\exp(4\phi) \bar{g}_{bc})) \quad (\text{eq27.104})$$

$$= \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} \partial_b (\exp(4\phi) \bar{g}_{ec}) + \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} \partial_c (\exp(4\phi) \bar{g}_{be}) - \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} \partial_e (\exp(4\phi) \bar{g}_{bc}) \quad (\text{eq27.105})$$

$$= \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} (\partial_b (\exp(4\phi)) \bar{g}_{ec} + \exp(4\phi) \partial_b \bar{g}_{ec}) + \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} (\partial_c (\exp(4\phi)) \bar{g}_{be} + \exp(4\phi) \partial_c \bar{g}_{be}) - \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} (\partial_e (\exp(4\phi)) \bar{g}_{bc} + \exp(4\phi) \partial_e \bar{g}_{bc}) \quad (\text{eq27.106})$$

$$= \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} (4 \exp(4\phi) \partial_b \phi \bar{g}_{ec} + \exp(4\phi) \partial_b \bar{g}_{ec}) + \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} (4 \exp(4\phi) \partial_c \phi \bar{g}_{be} + \exp(4\phi) \partial_c \bar{g}_{be}) - \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} (4 \exp(4\phi) \partial_e \phi \bar{g}_{bc} + \exp(4\phi) \partial_e \bar{g}_{bc}) \quad (\text{eq27.107})$$

$$= 2 \exp(-4\phi) \bar{g}^{ae} \exp(4\phi) \partial_b \phi \bar{g}_{ec} + \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} \exp(4\phi) \partial_b \bar{g}_{ec} + 2 \exp(-4\phi) \bar{g}^{ae} \exp(4\phi) \partial_c \phi \bar{g}_{be} + \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} \exp(4\phi) \partial_c \bar{g}_{be} - 2 \exp(-4\phi) \bar{g}^{ae} \exp(4\phi) \partial_e \phi \bar{g}_{bc} - \frac{1}{2} \exp(-4\phi) \bar{g}^{ae} \exp(4\phi) \partial_e \bar{g}_{bc} \quad (\text{eq27.108})$$

$$= 2 \bar{g}^{ae} \partial_b \phi \bar{g}_{ec} + \frac{1}{2} \bar{g}^{ae} \partial_b \bar{g}_{ec} + 2 \bar{g}^{ae} \partial_c \phi \bar{g}_{be} + \frac{1}{2} \bar{g}^{ae} \partial_c \bar{g}_{be} - 2 \bar{g}^{ae} \partial_e \phi \bar{g}_{bc} - \frac{1}{2} \bar{g}^{ae} \partial_e \bar{g}_{bc} \quad (\text{eq27.109})$$

$$= 2 \bar{g}^{ae} \partial_b \phi \bar{g}_{ec} + 2 \bar{g}^{ae} \partial_c \phi \bar{g}_{be} - 2 \bar{g}^{ae} \partial_e \phi \bar{g}_{bc} + \bar{\Gamma}^a_{bc} \quad (\text{eq27.110})$$

$$= 2 \bar{g}^a_c \partial_b \phi + 2 \bar{g}^{ae} \partial_c \phi \bar{g}_{be} - 2 \bar{g}^{ae} \partial_e \phi \bar{g}_{bc} + \bar{\Gamma}^a_{bc} \quad (\text{eq27.111})$$

$$= 2 \bar{g}^a_c \partial_b \phi + 2 \bar{g}^a_b \partial_c \phi - 2 \bar{g}^{ae} \partial_e \phi \bar{g}_{bc} + \bar{\Gamma}^a_{bc} \quad (\text{eq27.112})$$

```

1  # -----
2  # Check against prd67.
3
4  foo := @(Gamma).                # cdb(prd67.eq27.lcb,foo)
5  bah  = cdblib.get('prd67.eq27.rhs','prd67.json')  # cdb(prd67.eq27.prd,bah)
6
7  diff := @(foo) - @(bah).
8
9  distribute      (diff)
10 diff = product_sort (diff)
11 rename_dummies (diff)
12 map_sympy       (diff, "simplify")
13 canonicalise    (diff)          # cdb(prd67.eq27.chk,diff)

```

$$\text{prd67.eq27.lcb} := 2\bar{g}^a{}_c\partial_b\phi + 2\bar{g}^a{}_b\partial_c\phi - 2\bar{g}^{ae}\partial_e\phi\bar{g}_{bc} + \bar{\Gamma}^a{}_{bc}$$

$$\text{prd67.eq27.prd} := \bar{\Gamma}^a{}_{bc} + 2\bar{g}^a{}_c\partial_b\phi + 2\bar{g}^a{}_b\partial_c\phi - 2\bar{g}_{bc}\bar{g}^{ae}\partial_e\phi$$

$$\text{prd67.eq27.chk} := 0$$